Chapter 20
Quantum Nonlinear Oscillator

20.1 Perturbation Theory

Suppose we know eigenvalues and eigenstates of a Hamiltonian $\hat{H}_0$ and want to find them for a Hamiltonian $\hat{H} = \hat{H}_0 + \hat{V}$ in the form of series in $\hat{V}$ [18]. Let's concentrate on a non-degenerate eigenstate of the unperturbed Hamiltonian

$$\hat{H}_0 \, |\psi_0\rangle = E_0 \, |\psi_0\rangle \, .$$

After switching the perturbation on it transforms to a similar eigenstate of the full Hamiltonian

$$\hat{H} \, |\psi\rangle = E \, |\psi\rangle \, , \qquad E = E_0 + \delta E \, .$$

Let's normalize $|\psi\rangle$ in such a way that $\langle\psi_0|\psi\rangle = 1$, then $|\psi\rangle = |\psi_0\rangle + |\delta\psi\rangle$, $\langle\psi_0|\delta\psi\rangle = 0$. We need to solve the equation

$$\hat{H}_0 \, |\psi\rangle + \hat{V} \, |\psi\rangle = E \, |\psi\rangle \, .$$

Let's separate its components parallel and orthogonal to $|\psi_0\rangle$. The parallel part is singled out by multiplying by $\langle\psi_0|$:

$$\delta E = \langle\psi_0|\hat{V}|\psi\rangle \, .$$

The orthogonal part is singled out by the projector $\hat{P} = 1 - |\psi_0\rangle\langle\psi_0|$:

$$\hat{H}_0 \, |\delta\psi\rangle + \hat{P}\hat{V} \, |\psi\rangle = E \, |\delta\psi\rangle \, ,$$

or

$$|\delta\psi\rangle = \hat{D}\hat{V} \, |\psi\rangle \, , \qquad \hat{D} = \frac{\hat{P}}{E - \hat{H}_0} \, .$$

Solving this equation by iterations, we obtain

$$|\delta\psi\rangle = \hat{D}\hat{V} \, |\psi_0\rangle + \hat{D}\hat{V}\hat{D}\hat{V} \, |\psi_0\rangle + \cdots \, ,$$

A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics,
DOI 10.1007/978-3-319-00894-3_20,

and

$$\delta E = <\psi_0|\hat{V}|\psi_0> + <\psi_0|\hat{V}\hat{D}\hat{V}|\psi_0> + <\psi_0|\hat{V}\hat{D}\hat{V}\hat{D}\hat{V}|\psi_0> + \cdots .$$

Note that $\hat{D}$ contains $E = E_0 + \delta E$ and should be expanded in δE:

$$\hat{D} = \hat{G} - \hat{G}\delta E\hat{G} + \hat{G}\delta E\hat{G}\delta E\hat{G} - \cdots , \qquad \hat{G} = \frac{\hat{P}}{E_0 - \hat{H}_0} .$$

20.2 Nonlinear Oscillator

We are going to apply the perturbation theory to the nonlinear oscillator

$$\hat{H}_0 = \frac{\hat{p}^2 + \hat{x}^2}{2} , \qquad \hat{V} = \sum_{k=1}^{\infty} c_k \hat{x}^{k+2} .$$

The oscillation amplitude is ~ 1 if $n \sim 1$; therefore, $c_k \sim a^k$, where $a \sim 1/L \ll 1$, L is the characteristic length of the potential in the oscillator units. If $n \gg 1$, the amplitude is $\sqrt{n}$ times larger, and the real expansion parameter is $a\sqrt{n}$.

We concentrate on the eigenstate $|n>$ of $\hat{H}_0$ having the energy $E_0 = n + \frac{1}{2}$. In order to calculate δE up to the Mth order of the perturbation theory, we need to sum all expressions of the form

$$\left(\hat{V}_{k_N}\right)_{n,n+j_{N-1}} \hat{G}_{n+j_{N-1}} \left(\hat{\Delta}_{k_{N-1}}\right)_{n+j_{N-1},n+j_{N-2}} \hat{G}_{n+j_{N-2}} \cdots$$
$$\hat{G}_{n+j_2} \left(\hat{\Delta}_{k_2}\right)_{n+j_2,n+j_1} \hat{G}_{n+j_1} \left(\hat{V}_{k_1}\right)_{n+j_1,n} ,$$

where $\hat{\Delta}$ is $\hat{V}$ or $-\delta E$ and the sum of the orders of smallness $k_1 + k_2 + \cdots + k_N \leqslant M$; the sum over all nonzero $j_1,\ j_2,\ldots\ j_{N-1}$ is assumed. The following procedure prepares the values $V[k,j]$ of $\left(\hat{V}_k\right)_{n+j,n}$ ($x[k,j]$ means $\left(\hat{x}^k\right)_{n+j,n}$):

```
In[1] := Prepare[m_] := (M = m; x[1,1] = Sqrt[(n+1)/2]; x[1,-1] = Sqrt[n/2];
    x[1,0] = 0;
    Do[x[k,j] = If[j < k-1, (x[1,1]/.n -> n+j) * x[k-1,j+1],0]+
            If[j > 1-k, (x[1,-1]/.n -> n+j) * x[k-1,j-1],0],
        {k,2,m+2},{j,-k,k}];
    Do[V[k,j] = Simplify[c[k] * x[k+2,j]],{k,1,m},{j,-k-2,k+2,2}])
```

20.3 Energy Levels

The expressions we want to generate and calculate can be visualized as paths in the following graph:

```
In[2] := ParametricPlot[{{t,3 − 3*t},{t,−3+3*t},
    {t/2,3/2*t},{t/2,−3/2*t}},{t,0,1},
    PlotRange → {{0,1},{−3,3}}, AxesLabel → {l, j},
    Ticks → {{0,{1/2,M/2},{1,M}},
        {{−3,−3*M},{−3/2,−3/2*M},0,{3/2,3/2*M},{3,3*M}}}]
```

Out[2] =



Here l is the number of orders of smallness we need to distribute. We start at the point $l = M$, $j = 0$. At each step we consider all possible $\hat{\Delta}_k$. If we choose $(\hat{V}_k)_{n+j+i,n+j}$, we move k steps to the left and i steps vertically. If we choose $-\delta E_k$, we move k steps to the left horizontally (this choice is not allowed as the first or the last step). Whenever we hit the $j = 0$ axis, we have a complete expression for a contribution to δE. The fastest movement along j at varying l occurs when $\hat{V}_1$ is used, and its velocity is 3. Hence, in order to have enough time to return to $j = 0$, we should not leave the rhombus in the figure.

Suppose we have already generated a right-hand part a of the expression up to some $\hat{G}_{n+j}$ inclusively, and there remain l orders of smallness to distribute. The procedure considers all possible $(\hat{\Delta}_k)_{n+j+i,n+j}$ which can be inserted to the left of a. It may be $-\delta E_k$ with all possible values of k (if we are not at the very first step $l = M$), the maximum k is obtained from the rhombus. Or it may be $(\hat{V}_k)_{n+j+i,n+j}$ with all possible values of k and i. $\hat{V}_k = c_k \hat{x}^{k+2}$ has nonzero matrix elements for i from $-k-2$ to $k+2$ in steps of 2. The limits of the i loop follow from the intersection of this range with the rhombus, and the k loop terminates when this intersection disappears. If we happen to return to the initial state ($j + i = 0$), this means that the generation of an expression is complete, and it should be added to the element of the list d which accumulates contributions to δE (this contribution may contain lower-order $\delta E_k = \mathrm{de}[k]$). In all other cases, the procedure is called recursively, with l replaced by $l - k$, j by $j + i$, and a multiplied by $(\hat{\Delta}_k)_{n+j+i,n+j}$ and $\hat{G}_{n+j+i}$.

```
In[3] := v[l_, j_, a_] := (If[l == M, d = Table[0, {M}],
            Do[v[l − k, j, a ∗ de[k]/j], {k, 2, l − Abs[j]/3, 2}]];
        Do[If[j + i == 0, d[[M − l + k]]+ = a ∗ (V[k, i]/.n → n + j),
                v[l − k, j + i, −a ∗ (V[k, i]/.n → n + j)/(j + i)]],
            {k, Min[l, 1 + (3 ∗ l − Abs[j])/2]},
            {i, Max[−k − 2, −3 ∗ (l − k) − j], Min[k + 2, 3 ∗ (l − k) − j], 2}])
In[4] := Prepare[6]; v[M, 0, 1];
```

Now we substitute lower-order δE into expressions for higher-order ones, to get explicit formulas.

```
In[5] := Do[de[k] = Simplify[d[[k]]]; Print[Collect[de[k], c[_], Factor]], {k, 2, M, 2}]
```

$$\frac{1}{8}\left(-11-30n-30n^2\right)c[1]^2+\frac{3}{4}\left(1+2n+2n^2\right)c[2]$$

$$-\frac{15}{32}(1+2n)\left(31+47n+47n^2\right)c[1]^4+\frac{9}{8}(1+2n)\left(19+25n+25n^2\right)c[1]^2c[2]-\frac{1}{8}(1+2n)\left(21+17n+17n^2\right)c[2]^2-\frac{5}{8}(1+2n)\left(13+14n+14n^2\right)c[1]c[3]+\frac{5}{8}(1+2n)\left(3+2n+2n^2\right)c[4]$$

$$\frac{1}{128}\left(-39709-162405n-278160n^2-231510n^3-115755n^4\right)c[1]^6+\frac{3}{64}\left(15169+59385n+98160n^2+77550n^3+38775n^4\right)c[1]^4c[2]+\frac{3}{16}\left(111+347n+472n^2+250n^3+125n^4\right)c[2]^3+$$

$$\frac{1}{16}\left(-4517-16815n-26580n^2-19530n^3-9765n^4\right)c[1]^3c[3]+$$
$$\frac{1}{32}\left(-449-1400n-2030n^2-1260n^3-630n^4\right)c[3]^2-$$
$$\frac{15}{8}\left(12+35n+46n^2+22n^3+11n^4\right)c[2]c[4]+$$
$$c[1]^2\left(\frac{1}{32}\left(-11827-43479n-68424n^2-49890n^3-24945n^4\right)c[2]^2+\right.$$
$$\left.\frac{5}{16}\left(323+1125n+1668n^2+1086n^3+543n^4\right)c[4]\right)+$$
$$c[1]\left(\frac{3}{8}\left(474+1625n+2430n^2+1610n^3+805n^4\right)c[2]c[3]-\right.$$
$$\left.\frac{105}{16}\left(5+16n+22n^2+12n^3+6n^4\right)c[5]\right)+$$
$$\frac{35}{16}\left(3+8n+10n^2+4n^3+2n^4\right)c[6]$$

20.4 Correspondence Principle

At $n \gg 1$ the expansion parameter of the perturbation series is $a\sqrt{n}$ where $c_k \sim a^k$. Keeping only the highest powers of n in each order, we have

```
In[6] := Eq = Series[n+
        Sum[(Expand[(de[2*(j-1)]/.n -> 1/a)*a^j]/.a -> 0)*n^j,
            {j,2,M/2+1}],
    {n,0,M/2+1}]
```

$$\mathrm{Out}[6] = n+\left(-\frac{15}{4}c[1]^2+\frac{3c[2]}{2}\right)n^2+$$
$$\frac{1}{16}\left(-705c[1]^4+900c[1]^2c[2]-68c[2]^2-280c[1]c[3]+40c[4]\right)n^3-$$
$$\frac{5}{128}\left(23151c[1]^6-46530c[1]^4c[2]+19956c[1]^2c[2]^2-600c[2]^3+\right.$$
$$15624c[1]^3c[3]-7728c[1]c[2]c[3]+504c[3]^2-4344c[1]^2c[4]+$$
$$\left.528c[2]c[4]+1008c[1]c[5]-112c[6]\right)n^4+$$
$$O[n]^5$$

Bohr's correspondence principle must hold. From the quantum point of view, the particle at the nth energy level can radiate a photon, jumping to the $(n-1)$th one, or more generally to the $(n-k)$th one. The frequency of this photon is $E_n - E_{n-k}$, or approximately $\frac{\mathrm{d}E_n}{\mathrm{d}n}k$. From the classical point of view, the frequencies of the emitted light are equal to the oscillation frequency ω and its harmonics. Therefore, the oscillation frequency is

$$\omega = \frac{\mathrm{d}E_n}{\mathrm{d}n}\,.$$

In[7] := Wq = *D*[Eq, *n*]

$$\text{Out}[7] = 1+2\left(-\frac{15}{4}c[1]^2+\frac{3c[2]}{2}\right)n+$$
$$\frac{3}{16}\left(-705c[1]^4+900c[1]^2c[2]-68c[2]^2-280c[1]c[3]+40c[4]\right)n^2-$$
$$\frac{5}{32}\left(23151c[1]^6-46530c[1]^4c[2]+19956c[1]^2c[2]^2-600c[2]^3+\right.$$
$$15624c[1]^3c[3]-7728c[1]c[2]c[3]+504c[3]^2-4344c[1]^2c[4]+$$
$$\left.528c[2]c[4]+1008c[1]c[5]-112c[6]\right)n^3+$$
$$O[n]^4$$

We want to compare it with the result of the calculation in classical mechanics. But the quantum expression for ω is in terms of n and the classical one in terms of the oscillation amplitude a. We need to re-express both of them via the same quantity, the energy E.

In[8] := ne = InverseSeries[Eq, *e*]

$$\text{Out}[8] = e+\frac{3}{4}\left(5c[1]^2-2c[2]\right)e^2+$$
$$\frac{5}{16}\left(231c[1]^4-252c[1]^2c[2]+28c[2]^2+56c[1]c[3]-8c[4]\right)e^3+$$
$$\frac{35}{128}\left(7293c[1]^6-12870c[1]^4c[2]+5148c[1]^2c[2]^2-264c[2]^3+\right.$$
$$3432c[1]^3c[3]-1584c[1]c[2]c[3]+72c[3]^2-792c[1]^2c[4]+$$
$$\left.144c[2]c[4]+144c[1]c[5]-16c[6]\right)e^4+$$
$$O[e]^5$$

In[9] := Wqe = Simplify[Wq/. *n* → ne]

$$\text{Out}[9] = 1+\left(-\frac{15}{2}c[1]^2+3c[2]\right)e-$$
$$\frac{3}{16}\left(855c[1]^4-1020c[1]^2c[2]+92c[2]^2+280c[1]c[3]-40c[4]\right)e^2+$$
$$\frac{1}{32}\left(-164805c[1]^6+311670c[1]^4c[2]-94920c[1]^3c[3]-\right.$$
$$180c[1]^2\left(715c[2]^2-134c[4]\right)+5040c[1](9c[2]c[3]-c[5])+$$
$$\left.8\left(633c[2]^3-315c[3]^2-450c[2]c[4]+70c[6]\right)\right)e^3+$$
$$O[e]^4$$

Now we read the classical results for the energy Ec and the frequency Wc (written in terms of $A = a^2/2$, where the amplitude a is defined as the coefficient of the first harmonic $\cos\omega t$).

In[10] := <<class.m;

In[11] := Ec = Series[Ec, {*A*, 0, 3}]

$$\text{Out}[11] = A+\left(-\frac{37}{4}c[1]^2+\frac{9c[2]}{2}\right)A^2+$$
$$\frac{1}{192}\left(-9309c[1]^4+17796c[1]^2c[2]+300c[2]^2-10880c[1]c[3]+1920c[4]\right)A^3+$$
$$O[A]^4$$

In[12] := Wc = Series[Wc, {A, 0, 2}]

$$\text{Out}[12] = 1 + \left(-\frac{15}{2}c[1]^2 + 3c[2]\right)A - \frac{3}{16}\left(485c[1]^4 - 692c[1]^2c[2] + 20c[2]^2 + 280c[1]c[3] - 40c[4]\right)A^2 + O[A]^3$$

In[13] := Ae = Simplify[InverseSeries[Ec, e]]

$$\text{Out}[13] = e + \left(\frac{37c[1]^2}{4} - \frac{9c[2]}{2}\right)e^2 + \left(\frac{14055c[1]^4}{64} - \frac{4147}{16}c[1]^2c[2] + \frac{623c[2]^2}{16} + \frac{170}{3}c[1]c[3] - 10c[4]\right)e^3 + O[e]^4$$

In[14] := Wce = Simplify[Wc/.A → Ae]

$$\text{Out}[14] = 1 + \left(-\frac{15}{2}c[1]^2 + 3c[2]\right)e - \frac{3}{16}\left(855c[1]^4 - 1020c[1]^2c[2] + 92c[2]^2 + 280c[1]c[3] - 40c[4]\right)e^2 + O[e]^3$$

In[15] := Wqe − Wce

$$\text{Out}[15] = O[e]^3$$

20.5 States

The following procedure accumulates contributions to δE in elements of the list d and to $|\delta\psi>$ in the list dp. Now we have to consider the large triangle in the figure, not just the rhombus.

```
In[16] := v2[l_, j_, a_] := (dp[[M − l]]+ = a ∗ ket[n + j];
      If[l < M, Do[v2[l − k, j, a ∗ de[k]/j], {k, 2, l − 1, 2}]];
      Do[If[j + i == 0, d[[M − l + k]]+ = a ∗ (V[k, i]/.n → n + j),
               v2[l − k, j + i, −a ∗ (V[k, i]/.n → n + j)/(j + i)]],
         {k, l}, {i, −k − 2, k + 2, 2}])
In[17] := Prepare[2]; d = Table[0, {M}]; dp = Table[0, {M}];
      Clear[de]; v2[M, 0, 1];
In[18] := Do[de[k] = Simplify[d[[k]]];
      Print[Collect[dp[[k]], ket[_], Simplify]], {k, M}]
```

$$\frac{\sqrt{-2+n}\sqrt{-1+n}\sqrt{n}c[1]\text{ket}[-3+n]}{6\sqrt{2}} + \frac{3n^{3/2}c[1]\text{ket}[-1+n]}{2\sqrt{2}} - \frac{3(1+n)^{3/2}c[1]\text{ket}[1+n]}{2\sqrt{2}} - \frac{\sqrt{1+n}\sqrt{2+n}\sqrt{3+n}c[1]\text{ket}[3+n]}{6\sqrt{2}}$$

$$\frac{1}{144}\sqrt{-5+n}\sqrt{-4+n}\sqrt{-3+n}\sqrt{-2+n}\sqrt{-1+n}\sqrt{n}c[1]^2\text{ket}[-6+n] + \frac{1}{32}\sqrt{-3+n}\sqrt{-2+n}\sqrt{-1+n}\sqrt{n}\left((-3+4n)c[1]^2 + 2c[2]\right)\text{ket}[-4+n] +$$

$$\frac{1}{16}\sqrt{-1+n}\sqrt{n}\left(\left(1-19n+7n^2\right)c[1]^2+4(-1+2n)c[2]\right)\text{ket}[-2+n]+$$
$$\frac{1}{16}\sqrt{1+n}\sqrt{2+n}\left(\left(27+33n+7n^2\right)c[1]^2-4(3+2n)c[2]\right)\text{ket}[2+n]+$$
$$\frac{1}{32}\sqrt{1+n}\sqrt{2+n}\sqrt{3+n}\sqrt{4+n}\left((7+4n)c[1]^2-2c[2]\right)\text{ket}[4+n]+$$
$$\frac{1}{144}\sqrt{1+n}\sqrt{2+n}\sqrt{3+n}\sqrt{4+n}\sqrt{5+n}\sqrt{6+n}c[1]^2\text{ket}[6+n]$$

As an additional problem, calculate the average values of $\hat{x}^k$ over the states just obtained, for several k. At $n \gg 1$ compare them to the classical averages obtained from the particle's motion $x(t)$.